utilisation des développements en série limitée; utilisation d'un tableur
étude de la trajectoire d'un mobile par la méthode d'Euler

1.méthode d'Euler
1.1 principe de la méthode

Toute fonction f(t) infiniment dérivable dans son domaine de définition peut s'écrire sous la forme d'un développement de Taylor autour de $t = t_0$: f(t) = $f(t_0)+f'(t_0).\frac{(t-t_0)}{1!}+f''(t_0).\frac{(t-t_0)^2}{2!}+f'''(t_0).\frac{(t-t_0)^3}{3!}+...$

En choisissant $t-t_0$ suffisamment petit, on peut négliger les termes supérieurs à l'ordre n et déterminer de proche en proche la fonction f(t) si on connait $f(t_0)$, $f'(t_0)$, $f''(t_0)$, … ; ainsi, si on "discrétise" le temps en intervalles très petits, en définissant un "pas" $t_{n+1} - t_n = \tau$ l'expression précédente devient : $f(t_n+\tau)=f(t_n)+\left(\frac{\partial f}{\partial t}\right)_{t_n}\tau+\frac{1}{2!}\left(\frac{\partial^2 f}{\partial t^2}\right)_{t_n}\tau^2+\frac{1}{3!}\left(\frac{\partial^3 f}{\partial t^3}\right)_{t_n}\tau^3+...$

et si τ est très petit, on peut se limiter à l'ordre 2 : $f(t_n+\tau)=f(t_n)+\left(\frac{\partial f}{\partial t}\right)_{t_n}\tau+\frac{1}{2!}\left(\frac{\partial^2 f}{\partial t^2}\right)_{t_n}\tau^2$, ou même à l'ordre 1.

1.2 détermination d'une équation horaire

Supposons connues les conditions initiales d'un mouvement dont l'équation horaire à déterminer est x = x(t) et y = y(t).
A t_0, nous connaissons x_0, y_0, v_{x0}, v_{y0} et les forces appliquées donc les accélérations a_{x0} et a_{y0}.

développons à l'ordre 1 les composantes de la vitesse à un instant t_n : $v_x(t_n+\tau)=v_x(t_n)+\tau\left(\frac{\partial v_x}{\partial t}\right)_{t_n}=v_x(t_n)+\tau a_x(t_n)$

on voit que la vitesse à l'instant t_{n+1} s'obtient à partir de la vitesse et de l'accélération à l'instant t_n

et on calcule de proche en proche : $v_{x(n+1)} = v_{xn} + a_{xn}.\tau$, $v_{y(n+1)} = v_{yn} + a_{yn}.\tau$, $x_{n+1} = x_n + v_{xn}.\tau$ et $y_{n+1} = y_n + v_{yn}.\tau$

on peut donc connaître la position et la vitesse à un instant quelconque à partir de leurs valeurs à t = 0
remarque : pour la position, on peut pousser à l'ordre 2 puisqu'on connait l'accélération : $x_{n+1} = x_n + v_{xn}.\tau + \frac{1}{2}.a_{xn}.\tau^2$ etc..

Le théorème du centre d'inertie conduit à deux équation différentielle du second ordre : $m\frac{d^2x}{dt^2}=F_x(x,y)$ et

$m\frac{d^2y}{dt^2}=F_y(x,y)$ ce qui permet, le cas échéant, de recalculer forces et accélérations à chaque pas de calcul

L'utilisation d'un tableur permet d'effectuer rapidement un grand nombre de calculs, ce qui nous permettra de tracer la trajectoire, et de voir l'influence de divers paramètres.

	x	y	$v_x(x,y)$	$v_y(x,y)$	$a_x=F_x(x,y)/m$	$a_y=F_y(x,y)/m$
$t_0=0$	x_o	y_o	v_{xo}	v_{yo}	a_{x0}	a_{y0}
$t_1=\tau$	$x_1 = x_o + v_{xo}.\tau$	$y_1 = y_o + v_{yo}.\tau$	$v_{x1} = v_{xo} + a_{xo}.\tau$	$v_{y1} = v_{yo}+ a_{yo}.\tau$	a_x	a_y
t_n	...	...	...	...	...	...
t_{n+1}	$x_{n+1} = x_n + v_{xn}.\tau$	$y_{n+1} = y_n + v_{yn}.\tau$	$v_{x(n+1)} = v_{xn} + a_{xn}.\tau$	$v_{y(n+1)} = v_{yn} + a_{yn}.\tau$	a_x	a_y

2 étude du mouvement d'un mobile dans un champ de pesanteur uniforme
2.1 établissement des expressions :

Pour un masse m lancée vers le haut avec une vitesse initiale $\vec{v}_0$,
les équations de la dynamique donnent (ici pas de frottement):
$a_{xo} = 0$ et $a_{yo} = -g = cte$
connaissant v_{xo} et v_{yo} on utilisera la méthode précédente pour obtenir les équations horaires x(t) et y(t), et on comparera les résultats obtenus aux résultats théoriques :

$x(t) = x_o + v_{xo}t$

$y(t) = y_o + v_{yo}t - \frac{1}{2}gt^2$



2.2 utilisation du tableur (Excel)
ouvrir une feuille de calcul, et garnir les lignes **1** à **6** de la façon suivante, en respectant impérativement les numéros de ligne et de colonne. :

	A	B	C	D	E	F	G
1	trajectoire d'un projectile dans un champ de gravitation uniforme						
2							
3	tau	g		vox	voy		
4	0,08	-9,8		30	120		0
5							
6	t	x	y	vx	vy	ax	ay
7	0	0	0	30	120	0	-9,8

on définit ainsi les constantes et les valeurs initiales. Pour utiliser une constante dans un calcul, il faudra taper les références absolues de la cellule correspondante, avec le symbole $. par exemple, pour tau : $A$4 (On peut aussi attribuer un nom à la cellule, en rentrant ce nom dans le champ situé au dessus de la colonne A)

garnir ensuite la ligne **7** (valeurs initiales) : pour entrer une formule, on tape d'abord "=" puis la formule
en A7, B7, C7 : =0 en D7 : =D4 en E7 : =E4 en F7 : =0 en G7 : =B4

puis entrer les formules suivantes sur la ligne 8 :
A8 : =A7+$A$4
B8 : =B7+$A$4*D7
C8 : =C7+$A$4*E7
D8 : =D7+$A$4*F7
E8 : =E7+$A$4*G7
F8 : =F7
G8: =G7 (remarque : remplacer $A$4 par tau, si on a nommé la cellule)

on doit voir les valeurs ci-dessous sur la ligne 8:

6	t	x	y	vx	vy	ax	ay
7	0	0	0	30	120	0	-9,8
8	0,08	2,4	9,6	30	119,216	0	-9,8

faire un "copier-coller" de la ligne 8 sur 350 lignes environ : pour ce faire, sélectionner la ligne 8, cliquer sur le coin inférieur droit de la partie sélectionnée (où apparaît un petit carré), et "tirer" vers le bas sur environ 350 lignes. Le nombre exact importe peu.Les cellules sont calculées automatiquement; sinon activer la fonction "calcul".

2.3 tracé de la trajectoire

sélectionner les colonnes x et y jusqu'aux dernières valeurs numériques
ouvrir un graphique, choisir "nuages de points", lignes, et tracer le graphique sur une feuille annexe.On pourra ajuster quelques options de tracé, mais sans perdre de temps.
vérifier que la trajectoire obtenue a bien l'allure attendue :



exemple de graphe obtenu

2.4 trajectoire théorique

pour vérifier la validité du calcul précédent, vous allez tracer la trajectoire théorique à partir des équations horaires déduites des lois de la dynamique : $x(t) = x_o + v_{xo}t$ et $y(t) = y_o + v_{yo}t - \frac{1}{2}gt^2$

garnir le début des colonnes J,K,L comme ci-contre →

J	K	L
trajectoire théorique		
xth	yth	écart
0	0	0
2,4	9,56864	0,03136
4,8	19,07456	0,06272

puis entrer les formules suivantes :

en J7 : =$D$4*A7
en K7 : =$E$4*A7-0,5*9,8*A7^2
en L7 : =RACINE((J7-B7)^2+(K7-C7)^2)

(bien faire le lien avec les expressions théoriques)
l'écart représente la distance entre le point théorique et le point calculé

sélectionner J8,K8,L8, et tirer le coin droit pour calculer environ 350 points.

faire un clic droit sur le graphique précédent afin d'ajouter une nouvelle série de données constituée de $J$7:$J$350 pour xth, et $K$7:$K$350 pour yth ;
vérifier que la nouvelle courbe se superpose bien à la précédente. Que peut-on dire de la variation de l'écart au cours du temps ? Evaluer cet écart lorsque le mobile repasse en y = 0, et l'observer en modifiant les échelles pour agrandir le graphique au voisinage de ce point.

2.5 modification des paramètres :

pas de calcul : augmenter tau (par exemple 0,16 s) et valider pour recalculer la feuille. En cliquant sur les axes, changer les échelles du graphique de façon à observer la même portion d'espace que précédemment; conclure.
comparer les écarts au point d'impact pour trois ou quatre valeurs de tau, présenter les résultats dans un tableau, et conclure.

vitesse initiale : modifier les valeurs de v_{xo} et (ou) v_{xo}, et observer l'influence sur la trajectoire

calcul à l'ordre 2 : pour la position, on peut pousser à l'ordre 2 puisqu'on connait l'accélération :
$x_{n+1} = x_n + v_{xn}.\tau + \frac{1}{2}.a_{xn}.\tau^2$ etc.. compléter les expressions des lignes B8 et C8 avec le terme du second ordre :

en B8 : B7+$A$4*D7**+0,5*$A$4*$A$4*F7**
en C8 : C7+$A$4*E7**+0,5*$A$4*$A$4*G7**

comparer les écarts au point d'impact pour différentes valeurs de tau et conclure.



exemple de graphe obtenu

3. prise en compte d'une force de frottement.

la force de frottement exercée par l'air de masse volumique ρ sur un mobile lancé à la vitesse $\vec{v}$, et présentant un "maitre-couple" S (section interceptée par le solide sur un plan normal à la vitesse) est opposée à la vitesse, et a pour expression en régime turbulent : $\vec{F}_{fr} = -K_{fr}\rho S v^2 \vec{u}$



K_{fr} est un coefficient dépendant de la forme du corps

nous modéliserons donc cette force par un terme proportionnel à v², et opposé à $\vec{u}$, vecteur unitaire donnant la direction de la vitesse; les équations de la dynamique deviennent, en supposant S constante au cours du mouvement :

$ma_x = -K_{fr}\rho S v^2\, \vec{u}.\vec{u}_x$ et $ma_y = -mg - K_{fr}\rho S v^2\, \vec{u}.\vec{u}_y$

ou encore en posant $k = K_{fr}\rho S/m$ $\boxed{a_x = -kv^2 \dfrac{v_x}{\|\vec{v}\|} \quad \text{et} \quad a_y = -g - kv^2 \dfrac{v_y}{\|\vec{v}\|}}$

il faut donc rajouter ces termes dans le tableur.

ajouter **k** en G3, **0,0001** en G4 et **vcarre** en H6
entrer les expressions suivantes :

en F7 : =-$G$4*H7*(D7/RACINE(H7))
en G7 : =$B$4-$G$4*H7*(E7/RACINE(H7))
en H7 : =D7^2+E7^2

F	G	H
	k	
	0,0001	
ax	ay	vcarre

sélectionner F7,G7,H7, et tirer le coin droit vers le bas, pour calculer environ 350 points; vous devez voir les premières valeurs suivantes :

6	t	x	y	vx	vy	ax	ay	vcarre		xth	yth	écart
7	0	0	0	30	120	-0,37107951	-11,284318	15300		0	0	0
8	0,08	2,39881255	9,56389018	29,9703136	119,097255	-0,36806639	-11,2626372	15082,3757		2,4	9,56864	0,004896
9	0,16	4,79525982	19,0556301	29,9408683	118,196244	-0,36506757	-11,2411611	14866,8076		4,8	19,07456	0,01951436
10	0,24	7,18936107	28,4753579	29,9116629	117,296951	-0,3620829	-11,2198883	14653,2822		7,2	28,51776	0,04371643

Sur le graphique, doit apparaître maintenant la courbe théorique précédente, ne tenant pas compte du frottement, et la trajectoire calculée; comparer les positions du point d'impact.
Faire varier la valeur de k; qu'obtenez vous si celle-ci devient très importante ? le justifier théoriquement.



k= 0,0001



k=0,01